

Excel sayfasında iken **Alt+F11 tuşları ile İnsert/Module** ile aşağıdaki kod kopyalanıp yapıştırılır

```
Function ParaCevir(Para, Optional PBirim = "Lira", Optional KBirim = "Kuruş")
    Dim ParaStr As String
    Dim Lira As String, Kuruş As String

    If Not IsNumeric(Para) Then
        ParaCevir = "GİRİLEN DEĞER SAYI DEĞİL!"
        Exit Function
    End If

    ParaStr = Format(Abs(Para), "0.00")

    Lira = Left(ParaStr, Len(ParaStr) - 3)
    Kuruş = Right(ParaStr, 2)

    ParaCevir = IIf(Para < 0, "Eksi ", "") & Cevir(Lira) & " " & PBirim & " " & _
        IIf(Val(Kuruş) <> 0, Cevir(Kuruş) & " " & KBirim & " ", "")
End Function

Private Function Cevir(SayıStr As String) As String
    Dim Rakam(15)
    Dim c(3), Sonuc, e

    Birler = Array("", "bir", "iki", "üç", "dört", "beş", "altı", "yedi", "sekiz", "dokuz")
    Onlar = Array("", "on", "yirmi", "otuz", "kırk", "elli", "altmış", "yetmiş", "seksen", "doksan")
    Binler = Array("trilyon", "milyar", "milyon", "bin", "")


    SayıStr = String(15 - Len(SayıStr), "0") + SayıStr

    For i = 1 To 15
        Rakam(i) = Val(Mid$(SayıStr, i, 1))
    Next i

    Sonuc = ""
    For i = 0 To 4
        c(1) = Rakam(i * 3 + 1)
        c(2) = Rakam(i * 3 + 2)
        c(3) = Rakam(i * 3 + 3)
        If c(1) = 0 Then
            e = ""
        ElseIf c(1) = 1 Then
            e = "yüz"
        Else
            e = Birler(c(1)) + "yüz"
        End If
        e = e + Onlar(c(2)) + Birler(c(3))
        If e <> "" Then e = e + Binler(i)
        If (i = 3) And (e = "birbin") Then e = "bin"
        Sonuc = Sonuc + e
    Next i

    If Sonuc = "" Then Sonuc = "Sıfır"


    Cevir = UCase(Mid(Sonuc, 1, 1)) + Mid(Sonuc, 2, Len(Sonuc) - 1)
End Function
```



ve kaydet ile


Çalışma kitabını Excel Makro Etkin Çalışma kitabı(.....xlsm) olarak kaydettikten sonra, kod ekranı kapatılır.

Excel çalışma kitabına döndükten sonra güvenlik ayarlarının aşağıdaki gibi olmasını sağlarır.




Excel sayfasında Paraçevir işlevinin ekleneceği hücreye;

=ParaCevir(Hücre Adresi) yazılarak, çalıştırılır.



Eklenecek makro, bir eylemi gerçekleştirecekse (işlev/formül eklemeye gerek yoksa boş satır silme, sayfaları alfabetik sıralama gibi)

Geliştirici/Makrolar komutları ile açılan pencereden Makro bulunur ve Çalıştır ile eylemi gerçekleştirmesi sağlanır.



Yukarıdaki örnekte anlatıldığı gibi daha önce eklenen,

→ Sayfaları alfabetik sıralama

→ Sayfadaki gereksiz boşlukları silme makroları

İstenildiğinde bu pencereden bulunarak çalıştırılır.